Diferenciální rovnice II

Cílem tohoto kurzu je ukázat si různé příklady použití počítačového algebraického systému Maple při řešení obyčejných diferenciálních rovnic 2. řádu a soustav obyčejných diferenciálních rovnic. Příklady by nás měly přesvědčit o užitečnosti softwarových nástrojů jako je systém Maple na jedné straně a na druhé straně ujistit, že bez základních matematických znalostí nelze softwarové nástroje rozumně použít.

Poznámka. Zaměříme se na strukturovaný (přehledný) zápis příkazů v systému Maple.

Symbolicky diferenciální rovnici 2. řádu pro neznámou funkci $y = y(x)$ píšeme ve tvaru

$$F(x, y, y', y'') = 0\,.$$

Obecné řešení pro většinu rovnic neumíme najít. Zda systém Maple nalezne řešení, se přesvědčíme pomocí příkazu `dsolve(DR,y(x))`, kde první parametr `DR` je daná diferenciální rovnice a druhý `y(x)` je název pro hledanou funkci. Připomeňme si některé dovednosti, které jsme mohli získat v kurzu Derivace nebo Diferenciální rovnice I. V systému Maple derivaci funkce y podle x zapíšeme příkazem `diff(y(x),x)`. Podobně pro druhou derivaci funkce y podle x použijeme příkaz `diff(y(x),x$2)`.

Příklad 1. Nalezněte obecné řešení homogenní lineární diferenciální rovnice s konstantními koeficienty

$$y'' - 3y' + 2y = 0\,.$$

```
> DR:=diff(y(x),x$2)-3*diff(y(x),x)+2*y(x)=0;
```

$$DR := \left(\frac{\mathrm{d}^2}{\mathrm{d}x^2} y\left(x\right)\right) - 3\left(\frac{\mathrm{d}}{\mathrm{d}x} y\left(x\right)\right) + 2\,y\left(x\right) = 0$$

```
> dsolve(DR,y(x));
```

$$y(x) = \_C1\, e^{x} + \_C2\, e^{(2x)}$$

Příklad 2. Nalezněte obecné řešení nehomogenní lineární diferenciální rovnice s konstantními koeficienty

$$y'' - 3y' + 2y = x \sin x\,.$$

```
> DR:=diff(y(x),x$2)-3*diff(y(x),x)+2*y(x)=x*sin(x);
```

$$DR := \left(\frac{d^2}{dx^2} y(x)\right) - 3\left(\frac{d}{dx} y(x)\right) + 2\, y(x) = x \sin(x)$$

```
> dsolve(DR,y(x));
```

$$y(x) = \frac{3}{10}\, x \cos(x) + \frac{1}{10}\, x \sin(x) - \frac{3}{25} \sin(x) + \frac{17}{50} \cos(x) + e^{(2x)}\_C1 + e^{x}\_C2$$

Příklad 3. Nalezněte obecné řešení homogenní lineární diferenciální rovnice

$$x^2 y'' - x y' + y = 0\,.$$

```
> DR:=x^2*diff(y(x),x$2)-x*diff(y(x),x)+y(x)=0;
```

$$DR := x^2 \left(\frac{d^2}{dx^2} y(x)\right) - x\left(\frac{d}{dx} y(x)\right) + y(x) = 0$$

```
> dsolve(DR,y(x));
```

$$y(x) = \_C1\, x + \_C2\, x \ln(x)$$

Příklad 4. Nalezněte obecné řešení nehomogenní lineární diferenciální rovnice

$$x^2 y'' - xy' + y = x \ln x\,.$$

```
> DR:=x^2*diff(y(x),x$2)-x*diff(y(x),x)+y(x)=x*ln(x);
```

$$DR := x^2 \left(\frac{\mathrm{d}^2}{\mathrm{d}x^2} y(x)\right) - x \left(\frac{\mathrm{d}}{\mathrm{d}x} y(x)\right) + y(x) = x \ln(x)$$

```
> dsolve(DR,y(x));
```

$$y(x) = x\_C2 + x \ln(x)\, \_C1 + \frac{1}{6} (\ln(x))^3\, x$$

Poznámka. Systém Maple označuje, pro nás trochu netradičně, integrační konstanty $\_C1$, $\_C2$. Cílem příkladů 1–4 je ukázat, že systém Maple nalezne řešení některých diferenciálních rovnic mnohem rychleji než lidské bytosti. Řešení příkladů 1–4 umíme sami nalézt.

Zkouška správnosti řešení obyčejné diferenciální rovnice se v systému Maple provede pomocí příkazu `odetest(res,DR)` , kde `res` je rovnice, která definuje jistou funkci buď analyticky nebo implicitně a o které se chceme přesvědčit, že je řešením diferenciální rovnice `DR`. Je-li testovaná funkce řešením diferenciální rovnice `DR`, pak systém Maple vrátí hodnotu 0. Vraťme se k příkladu jedna a proveďme zkoušku.

```
> res:=dsolve(DR,y(x));
```

$$res := y(x) = \_C1\, e^{x} + \_C2\, e^{(2x)}$$

```
> odetest(res,DR);
```

$$0$$

Příklad 5. Nalezněte partikulární řešení homogenní lineární diferenciální rovnice s počáteční podmínkou

$$(e^x + 1)y'' - 2y' - e^x y = 0\,, \quad y(0) = 1\,, \quad y'(0) = 0\,.$$

Určete jeho definiční obor a ověřte, že toto partikulární řešení nabývá pro $x_0 = 0$ lokální minimum. Nakreslete graf řešení.

```
> DR:=(exp(x)+1)*diff(y(x),x$2)-2*diff(y(x),x)-exp(x)*y(x)=0;
```

$$DR := (e^x + 1)\left(\frac{d^2}{dx^2} y(x)\right) - 2\left(\frac{d}{dx} y(x)\right) - e^x y(x) = 0$$

```
> PP:=y(0)=1,D(y)(0)=0;
```

$$PP := y(0) = 1, \mathrm{D}(y)(0) = 0$$

```
> res:=dsolve({DR,PP},y(x));
```

$$res := y(x) = \frac{3}{2}\,\frac{1}{e^x + 1} + \frac{1}{2}\,\frac{(e^x)^2}{e^x + 1}$$

```
> f:=unapply(rhs(combine(res)),x);
```

$$f := x \mapsto \frac{3}{2}\,\frac{1}{e^x+1} + \frac{1}{2}\,\frac{e^{(2x)}}{e^x+1}$$

```
> f(0);
```

$$1$$

```
> D(f)(0);
```

$$0$$

```
> D(D(f))(0);
```

$$\frac{1}{2}$$

```
> limit(f(x),x=-infinity);
```

$$\frac{3}{2}$$

```
> limit(f(x),x=infinity);
```

$$\infty$$

```
> plot(f(x),x=-2..2,y=0..1.5);
```



Poznámka. Z přednášek z MII víme, že každá lineární diferenciální rovnice s počáteční podmínkou má řešení, které je jednoznačné. Dále víme, jak určit definiční obor řešení. Vzhledem k tomu, že koeficienty $a_0(x) = e^x + 1$, $a_1(x) = -2$, $a_2(x) = -e^x$ jsou spojité funkce na $\mathbb{R}$ a že $a_0(x) \neq 0$ pro $\forall x \in \mathbb{R}$, definiční obor partikulárního řešení je $\mathbb{R}$. Tedy definiční obor řešení musíme určit na základě znalostí z teorie lineárních diferenciálních rovnic. Výpočty limit, které jsme provedli v průběhu řešení, napomáhají jisté představě o grafu řešení.

Poznámka. Nyní si vysvětlíme (v některých případech možná jen připomeneme) nové příkazy, které jsme použili při řešení Příkladu 5. Chceme-li najít řešení počáteční úlohy, pak použijeme příkaz `dsolve`, kde vedle diferenciální rovnice oddělené čárkou napíšeme počáteční podmínku a uzavřeme do složených závorek. Příkaz `dsolve` vrací řešení ve tvaru rovnice, na jejíž pravé straně se obvykle nachází funkční předpis daného řešení. Potřebujeme-li získat tento předpis, použijeme příkaz `rhs` (right-hand side). Protože Příklad 5 vyžaduje práci s řešením jako funkcí, musíme ji vytvořit. K tomu slouží příkaz `unapply(vyraz,x)` , který z výrazu `vyraz` udělá funkci proměnné `x`.

Ve Sbírce příkladů z matematiky se můžeme setkat s úlohou, dokázat, že dané funkce y_1, y_2 tvoří v intervalu I fundamentální systém řešení homogenní lineární diferenciální rovnice 2. řádu, rovnici sestavit a napsat její obecné řešení. Víme, že nejprve musíme ověřit, že $W_{y_1,y_2}(x) \neq 0$ alespoň pro jedno $x \in I$, kde W_{y_1,y_2} je Wronského determinant (Wronskián) funkcí y_1, y_2. Pomocí příkazu `wronskian(seznam,x)` vytvoříme matici, kde na (i,j)-té pozici je $(i-1)$-vá derivace j-té funkce ze seznamu (`seznam`) podle proměnné `x`. Seznam prvků vytvoříme tak, že tyto prvky oddělené čárkou uzavřeme do hranatých závorek. Tak stanovíme jejich pořadí. Determinant matice spočteme užitím příkazu `det` . Připomeňme, že příkazy lineární algebry jsou součástí souboru programů `linalg`, který musíme přivolat příkazem `with(linalg):` .

Příklad 6. Dokažte, že funkce $y_1 = e^{-2x}$ a $y_2 = xe^{-2x}$ tvoří v $\mathbb{R}$ fundamentální systém řešení homogenní lineární diferenciální rovnice 2. řádu, rovnici sestavte a napište její obecné řešení.

```
> with(linalg):
```

```
Warning, the protected names norm and trace have been cervenaefined and unprotected
```

```
> fs:=[exp(-2*x),x*exp(-2*x)];
```

$$fs := [e^{(-2x)}, x\, e^{(-2x)}]$$

```
> Wr:=wronskian(fs,x);
```

$$Wr := \begin{bmatrix} e^{(-2\,x)} & xe^{(-2\,x)} \\ -2\,e^{(-2\,x)} & e^{(-2\,x)} - 2\,xe^{(-2\,x)} \end{bmatrix}$$

```
> det(Wr);
```

$$\left(e^{(-2\,x)}\right)^2$$

```
> sez:=[y(x),exp(-2*x),x*exp(-2*x)];
```

$$sez := [y(x), e^{(-2\,x)}, x\,e^{(-2\,x)}]$$

```
> W:=wronskian(sez,x);
```

$$W := \begin{bmatrix} y(x) & e^{(-2\,x)} & xe^{(-2\,x)} \\ \frac{\mathrm{d}}{\mathrm{d}x}y(x) & -2\,e^{(-2\,x)} & e^{(-2\,x)} - 2\,xe^{(-2\,x)} \\ \frac{\mathrm{d}^2}{\mathrm{d}x^2}y(x) & 4\,e^{(-2\,x)} & -4\,e^{(-2\,x)} + 4\,xe^{(-2\,x)} \end{bmatrix}$$

```
> det(W)=0;
```

$$4\,y(x)\left(e^{(-2\,x)}\right)^2 + 4\left(\frac{\mathrm{d}}{\mathrm{d}x}y(x)\right)\left(e^{(-2\,x)}\right)^2 + \left(\frac{\mathrm{d}^2}{\mathrm{d}x^2}y(x)\right)\left(e^{(-2\,x)}\right)^2 = 0$$

```
> simplify(%);
```

$$e^{(-4\,x)}\left(4\,y(x) + 4\,\frac{\mathrm{d}}{\mathrm{d}x}y(x) + \frac{\mathrm{d}^2}{\mathrm{d}x^2}y(x)\right) = 0$$

Neboť platí, že $W_{y_1,y_2}(x) = e^{-4x} \neq 0\,, \forall x \in \mathbb{R}\,,$ jsou funkce $y_1 = e^{-2x}$ a $y_2 = xe^{-2x}$ lineárně nezávislé a tedy tvoří v $\mathbb{R}$ fundamentální systém řešení nějaké homogenní lineární diferenciální rovnice 2. řádu, např. námi sestavené diferenciální rovnice $y'' + 4y' + 4y = 0\,.$ Obecné řešení této rovnice má tvar $y(x) = C_1\,y_1 + C_2\,y_2\,,\; x \in \mathbb{R}\,,\; C_1, C_2 \in \mathbb{R}\,.$

Poznámka. Ze vztahu pro obecné řešení homogenní lineární diferenciální rovnice 2. řádu vidíme, že funkce y, y_1, y_2 jsou lineárně závislé, neboť y je lineární kombinací $y_1, y_2\,.$ Tedy musí platit, že $W_{y,y_1,y_2}(x) = 0\,, \forall x \in \mathbb{R}\,.$ Tento vzoreček jsme využili pro sestavení diferenciální rovnice, jejíž řešení jsou zadané funkce.

Příklad 7. Určete v intervalu $x \in (0, \pi)$ reálný fundamentální systém řešení homogenní lineární diferenciální rovnice s nekonstantními koeficienty

$$y'' - \left(\frac{2}{x} + \operatorname{cotg} x\right) y' + \left(\frac{2}{x^2} + \frac{\operatorname{cotg} x}{x}\right) y = 0\,.$$

```
> DR:= diff (y(x),x$2)-(2/x+cot(x))*diff(y(x),x)+
                                   +(2/(x^2)+cot(x)/(x))*y(x)=0;
```

$$DR := \left(\frac{d^2}{dx^2}y(x)\right) - \left(\frac{2}{x} + \cot(x)\right)\left(\frac{d}{dx}y(x)\right) + \left(\frac{2}{x^2} + \frac{\cot(x)}{x}\right) y(x) = 0$$

```
> res:=dsolve(DR,y(x));
```

$$res := y(x) = \frac{\_C1\, x\sqrt{\sin(x)}}{((1+\cos(x))\,(-1+\cos(x)))^{(\frac{1}{4})}} + \frac{\_C2\, x\sqrt{\sin(x)}\cos(x)}{((1+\cos(x))\,(-1+\cos(x)))^{(\frac{1}{4})}}$$

```
> simplify(res);
```

$$y(x) = \frac{x\,\sqrt{\sin(x)}(\_C1 + \_C2\ \cos(x))}{(-1+\cos(x)^2)^{\left(\frac{1}{4}\right)}}$$

Poznámka. Systém Maple nám dal řešení (res), které jsme se pokusili zjednodušit pomocí příkazu `simplify(res);`. V tomto případě i upravené řešení dává pouze komplexní fundamentální systém, neboť pro $x \in (0,\pi)$ platí, že

$$-1+\cos^2(x) < 0 \;\Rightarrow\; (-1+\cos^2(x))^{\frac{1}{4}} \quad \text{je funkční předpis komplexní funkce reálné proměnné}\,.$$

Je na nás, abychom si povšimli, že pro $x \in (0,\pi)$ je

$$\frac{\sqrt{\sin(x)}}{(-1+\cos^2(x))^{\frac{1}{4}}} = \left(\frac{1}{-1}\right)^{\frac{1}{4}} \frac{\sqrt{\sin(x)}}{\sqrt[4]{1-\cos^2(x)}} = (-1)^{\frac{1}{4}}\,\frac{\sqrt{\sin(x)}}{\sqrt{\sin(x)}} = \sqrt[4]{-1}\,.$$

Tedy

$$y(x) = \sqrt[4]{-1}\,(C_1\,x + C_2\,x\cos(x))\,.$$

Odtud je vidět, že $y_1(x) = x$ a $y_2(x) = x\cos(x)$. Dalším výpočtem se přesvědčíme, že y_1 a y_2 jsou dvě lineárně nezávislá řešení zadané rovnice, tudíž tvoří fundamentální systém.

```
> vys:=y(x)=C1*x+C2*x*cos(x);
```

$$vys := y\,(x) = \_C1\;x + \_C2\;x\cos\,(x)$$

```
> odetest(vys,DR);
```

```
> with(linalg):
```

```
Warning, the protected names norm and trace have been cervenaefined and unprotected
```

```
> fs:=[x,x*cos(x)];
```

$$fs := [x, x\cos(x)]$$

```
> Wr:=wronskian(fs,x);
```

$$Wr := \begin{bmatrix} x & x\cos(x) \\ 1 & \cos(x) - x\sin(x) \end{bmatrix}$$

```
> det(Wr);
```

$$-\sin(x)\,x^2$$

Množina funkcí $\{x\,, x\cos x\}$ tvoří v intervalu $(0,\pi)$ reálný fundamentální systém řešení, protože tyto funkce jsou řešením zadané diferenciální rovnice a jsou lineárně nezávislé $(\,-x^2\sin(x)\neq 0\,,\ \forall x\in(0,\pi)\,)$.

V další části se zabýváme homogenní soustavou lineárních diferenciálních rovnic 1. řádu s konstantními koeficienty

$$\begin{aligned} x' &= a_{11}x + a_{12}y \\ y' &= a_{21}x + a_{22}y\,, \end{aligned} \tag{1}$$

kde $a_{ij} \in \mathbb{R}\,, i, j = 1, 2\,.$ V kurzu Matematika II jsme se naučili soustavu (1) řešit, má-li matice soustavy (1) dvě různá reálná vlastní čísla nebo dvě imaginární (komplexně sdružená) vlastní čísla. Systém Maple může provést výpočet za nás pro jakoukoliv soustavu (1). Stačí užít již známý příkaz

$$\texttt{dsolve}(\{\texttt{DR1}, \texttt{DR2}, \texttt{PP}\}, \{\texttt{x(t)}, \texttt{y(t)}\});\,,$$

kde `DR1` a `DR2` jsou dané diferenciální rovnice, `PP` je počáteční podmínka, je-li zadaná, a `x(t),y(t)` jsou symboly pro závisle proměnné. Grafy složek řešení a trajektorie snadno získáme užitím příkazu

$$\texttt{DEplot}([\texttt{DR1}, \texttt{DR2}], [\texttt{x(t)}, \texttt{y(t)}], \texttt{t} = \texttt{a..b}, [[\texttt{PP}]], \texttt{volba});\,,$$

kde `t=a..b` je rozmezí nezávisle proměnné. Chceme-li nakreslit graf první složky řešení $x(t)$ za `volba` dosadíme `scene=[t,x]`. Podobně vytvoříme graf druhé složky $y(t)\,.$ V případě, že kreslíme trajektorii, použijeme `scene=[x,y]` nebo nic. Pozor! Příkaz `DEplot` je součástí souboru programů `DEtools`. Buď tento soubor programů před použitím příkazu `DEplot` přivoláme příkazem `with(DEtools):` nebo musíme užít komplikovanější tvar `DEtools[DEplot](...);`. S podobným upozorněním jsme se již setkali, když jsme používali příkazy lineární algebry.

Poznámka. Parametry příkazu `DEplot` nemusí být seznamy objektů. V příkazu lze též užít množiny objektů (složené závorky). Tato volnost neplatí například pro příkaz `dsolve`. V jeho případě je nutností užít množinu objektů. Tento fakt nepovažujeme za zásadní, abychom si ho museli pamatovat. Rozumnější je, podívat se do nápovědy (Help) nebo postupovat pokus omyl.

Příklad 8. Najděte obecné řešení autonomní homogenní soustavy lineárních diferenciálních rovnic

$$\begin{aligned} x' &= -3x + \sqrt{2}y \\ y' &= \sqrt{2}x - 2y\,. \end{aligned}$$

Dále nalezněte partikulární řešení, které vyhovuje počáteční podmínce

$$x(0) = 4\,, \quad y(0) = \sqrt{2}\,.$$

Nakreslete trajektorii, která prochází bodem $(4, \sqrt{2})$, a grafy jednotlivých složek nalezeného partikulárního řešení.

```
> with(DEtools):
```

```
> DR1:=diff(x(t),t)=-3*x(t)+sqrt(2)*y(t);
```

$$DR1 := \frac{\mathrm{d}}{\mathrm{d}t}x(t) = -3\,x(t) + \sqrt{2}y(t)$$

```
> DR2:=diff(y(t),t)=sqrt(2)*x(t)-2*y(t);
```

$$DR2 := \frac{\mathrm{d}}{\mathrm{d}t}y(t) = \sqrt{2}\,x(t) - 2y(t)$$

```
> dsolve({DR1,DR2},{x(t),y(t)});
```

$$\left\{y(t) = -\frac{1}{2}\,\left(\_C1\,e^{(-4\,t)} - 2\,\_C2\,e^{(-t)}\right)\sqrt{2}, x(t) = \_C1\,e^{(-4\,t)} + \_C2\,e^{(-t)}\right\}$$

```
> PP:=x(0)=4,y(0)=sqrt(2);
```

$$PP := x(0) = 4, y(0) = \sqrt{2}$$

```
> dsolve({DR1,DR2,PP},{x(t),y(t)});
```

$$\left\{x(t) = 2\,e^{(-4\,t)} + 2\,e^{(-t)}, y(t) = -\frac{1}{2}\,\left(2\,e^{(-4\,t)} - 4\,e^{(-t)}\right)\sqrt{2}\right\}$$

```
> DEplot([DR1,DR2],[x(t),y(t)],t=0..2,[[x(0)=4,y(0)=sqrt(2)]],
  x=-4..4,y=-2..2);
```




```
> DEplot([DR1,DR2],[x(t),y(t)],t=-0.2..2,[x(0)=4,y(0)=sqrt(2)],
  scene=[t,x]);
```




```
> DEplot([DR1,DR2],[x(t),y(t)],t=-0.2..2,[x(0)=4,y(0)=sqrt(2)],
  scene=[t,y],stepsize=0.05);
```



Poznámka. Při kreslení trajektorie explicitně zadáváme rozmezí x-ové a y-ové souřadnice. U grafu druhé složky řešení používáme volbu `stepsize`, která nám umožňuje zjemnit krok numerické integrace, a tak vyhladit graf.

Obecné řešení soustavy (1) je popsáno vztahem

$$\begin{pmatrix} x(t) \\ y(t) \end{pmatrix} = c_1 e^{-4t} \begin{pmatrix} 1 \\ -\frac{\sqrt{2}}{2} \end{pmatrix} + c_2 e^{-t} \begin{pmatrix} 1 \\ \sqrt{2} \end{pmatrix}$$

a parametrické rovnice trajektorie, která prochází bodem $(4, \sqrt{2})$, mají tvar

$$\begin{aligned} x(t) &= 2e^{-4t} + 2e^{-t} \\ y(t) &= -\sqrt{2}e^{-4t} + 2\sqrt{2}e^{-t} \,. \end{aligned}$$

Spočteme-li limity $\lim_{t\to\infty} x(t)$ a $\lim_{t\to\infty} y(t)$, zjistíme, že trajektorie vchází do rovnovážného stavu $(0,0)$, což odpovídá obrázku.

Poznámka. Povšimněme si, že výstupem příkazu `DEplot` je kromě trajektorie také vektorové pole soustavy (1). Z obrázku tedy získáme jistou představu o všech trajektoriích soustavy (1). Podívejme se na některé trajektorie, které můžeme snadno získat z obecného řešení. Nechť například $c_1 = 0$. Pak $y(t) = \sqrt{2}x(t)$. Vzhledem k tomu, že $(0,0)$ je jediný rovnovážný stav soustavy, jsou polopřímky $y = \sqrt{2}x\,, x > 0$ a $y = \sqrt{2}x\,, x < 0$ dvě trajektorie. Podobně dostaneme dvě polopřímky jako dvě trajektorie soustavy (1), volíme-li $c_2 = 0$.